[Android Software/Hacking General [Developers Only]] atomicrabbit:
I created this vbscript a few years ago for the same reason. Works well. I have it set up as a Rule in Outlook, so whenever I get a trunk outage email from PBXes, it runs this script automatically.

dslreports.com/forum/r24908015-PBXes-free-acount-trunk-outtage-quick-fix

I've pasted the script here just in case the dsl reports post gets removed:

'###########################################
'## PBXes Trunk Outtage Recovery
'## for Free Users
'##
'## Author: atomicrabbit (atombomb2k@hotmail.com)
'## Date: 2010.10.08
'###########################################

'Edit the following variables appropriately
'------- START EDITING HERE ... DO NOT TOUCH ANYTHING BELOW -------
'Setting Page
' Set this to one of the settigs pages of PBXes
pbxes_settingPage = "https://www4.pbxes.com/config.php?display=extensions&extdisplay=100"

'Form Name
' Set this to the form name on the settings page specified above.
' It MUST be relative to the form on the page specified above
pbxes_formName = "addNew"

'PBXes Login Info
pbxes_username = "your_username" 'Your PBXes.com username
pbxes_password = "your_password" 'Your PBXes.com passowrd

'IE Window Visibility
' True - IE window will be visible while processing the code
' False - IE window will be hidden and run the code in the background
pbxes_windowVisible = False

'Browser Timeout
' Set max timeout for browser responsiveness (in seconds)
' Default: 7
pbxes_timeout = 7

'------- STOP EDITING HERE ... DO NOT TOUCH ANYTHING BELOW -------

pbxes_applySettings = pbxes_settingPage + "&clk_reload=true"
pbxes_timeout = pbxes_timeout * 5 'because the iterations are 200ms
pbxes_tCount = 1

'Create IE object
Set objIE = CreateObject("InternetExplorer.Application")
set WshShell = WScript.CreateObject("WScript.Shell")

'Go to PBXes.com Extensions page
objIE.Navigate pbxes_settingPage
objIE.Visible = pbxes_windowVisible
Wscript.Sleep 1000

'Check if not logged in
if not objIE.Document.Forms("login") is nothing then
' Login to PBXes if you're not already
objIE.Document.All.Item("username").Value = pbxes_username
objIE.Document.All.Item("password").Value = pbxes_password
objIE.Document.All.Item("remember").checked = "false"
objIE.Document.All.Item("btnSubmit").click
Wscript.Sleep 2000

'Go to settings page
objIE.Navigate pbxes_settingPage

'Wait for page to load
Do
Wscript.Sleep 200
pbxes_tCount = pbxes_tCount + 1
Loop While objIE.Document.Forms(pbxes_formName) is nothing AND pbxes_tCount < pbxes_timeout
end if

if pbxes_tCount >= pbxes_timeout AND objIE.Document.Forms(pbxes_formName) is nothing Then

MsgBox "Browser timeout. Try running thescript again.", vbOkOnly, "PBXes Trunk Outtage"
'exit script

Elseif not objIE.Document.Forms(pbxes_formName) is nothing Then

'Submit form
call objIE.Document.Forms(pbxes_formName).elements("Submit").click
Wscript.Sleep 2000

'Apply the changes (red bar at top)
objIE.Navigate pbxes_applySettings
Wscript.Sleep 2000

'Close IE
objIE.Quit

MsgBox "Recovered from trunk outtage!", vbOkOnly, "PBXes Trunk Outtage"

Else
MsgBox "Could not find '" & pbxes_formName & "' form on '" & pbxes_settingPage & "'.", vbOkOnly, "PBXes Trunk Outtage"
'exit script
End if

'Clear variables
Set objIE = Nothing
Set pbxes_settingPage = Nothing
Set pbxes_applySettings = Nothing
Set pbxes_username = Nothing
Set pbxes_password = Nothing
Set pbxes_windowVisible = Nothing
Set pbxes_timeout= Nothing
Set pbxes_tCount= Nothing

Wscript.quit